

Mathematik mit Mathematica

*Praktikum im Wintersemester 2021/22 an der TU Braunschweig
betreut von Prof. Dr. Michael Herrmann*

Lösungen Serie 04

Aufgabe 1

```
In[ ]:= (* definiere die Kurve *)  
alpha[t_] = 2 Cos[2 * t] * {Cos[t], Sin[t]};  
(* berechne normierten Tangentenvektor *)  
e1[t_] = alpha'[t] / Norm[alpha'[t]];  
(* berechne normierten Normalenvektor *)  
e2[t_] = {-e1[t][[2]], e1[t][[1]]};
```


```

In[ ]:= (* erzeuge Liste von Bildern *)
frames =
  Table[
    (* plote die Kurve *)
    ParametricPlot[
      alpha[t],
      {t, 0, 2 Pi}, (* Laufvariable der Kurve in jedem Bild *)
      (* ein paar Parameter, damit es hübsch aussieht *)
      Axes → None,
      PlotStyle → {Gray, Thickness[0.01]},
      ImageSize → 72 × 4,
      PlotLabel →
        Style[
          "Zweibein auf Kleeblatt",
          FontFamily → "Bradley Hand",
          FontSize → 24
        ],
      PlotRange → {{-2.6, 2.6}, {-2.6, 2.6}},
      (* male die Vektoren in ausgezeichnetem Punkt *)
      Epilog → {
        Thickness[0.01], Arrowheads[0.1],
        Darker[Green], Arrow[{alpha[s], alpha[s] + e1[s]}],
        Darker[Blue], Arrow[{alpha[s], alpha[s] + e2[s]}],
        Gray, PointSize[.03], Point[alpha[s]]
      }
    ],
    {s, 0, 2 * Pi, 2 * Pi / 200} (* Laufvariable für die verschiedenen Bilder *)
  ];

```

```
In[ ]:= (* animiere Liste von Bildern *)  
ListAnimate[frames]
```

Out[]:=



```
In[ ]:= (* erzeuge Filmchen zum Verschenken *)  
SetDirectory["~/Desktop"];  
Export["frenet.avi", frames]
```


Out[]:= frenet.avi

Aufgabe 2

```

In[ ]:= Manipulate[
  ParametricPlot[
    Exp[-sigma * t] * {Cos[omega * t], Sin[omega * t]}, (* das ist die Kurve *)
    {t, 0, 10 Pi},
    (* ein paar Optionen *)
    PlotRange -> {{-1.3, 1.3}, {-1.3, 1.3}},
    ImageSize -> 72 * 4,
    PlotStyle -> {Darker[Red], Thickness[0.02]},
    (* der Text als Graphik *)
    Epilog ->
      {Text[Style["Eine Spirale",
        FontFamily -> "Bradley Hand", FontSize -> 24], {- .6, 1.}]}
  ],
  (* die Parameter für Manipulate *)
  {{omega, 1, "Winkelgeschwindigkeit"}, -3, 3},
  {{sigma, .15, "Dämpfung"}, -.2, .5}
]

```



Aufgabe 3

```
In[ ]:= Manipulate[
  MatrixPlot[
    (* erzeugt Matrix von Zufallszahlen mit vorgegebener Verteilung *)
    RandomVariate[
      BernoulliDistribution[p], (* das ist die Verteilung *)
      {20, 20}
      (* Dimension des Ergebnisses, hier 20x20 Matrix *)
    ],
    (* setze Optionen *)
    ColorFunction -> "Monochrome",
    FrameTicks -> None,
    ImageSize -> 72 x 4
  ],
  {{p, 0.5, "Bernoulli-Parameter"}, 0, 1}
]
```

Out[]:=

